CONVOLUTION

PURPOSE

Compute the numerical convolution of two variables.

DESCRIPTION

Mathematically, the convolution of 2 continuous distributions g and h is defined as:

$$g*h = \int_{-\infty}^{\infty} g(\tau)h(t-\tau)d\tau \qquad \textbf{(EQ 3-29)}$$

In practice, h is typically a data stream while g is a response function. The response function is typically a peaked function that goes to zero in both directions from that peak. The effect of convolution is to smear the data stream with the response function.

DATAPLOT computes the convolution from the functions sampled at discrete points (see the sample program for an example of how to evaluate a function at a discrete set of points). This is referred to as discrete convolution. If X is the data stream with n_x points and Y is the response function with n_y points, then DATAPLOT computes the convolution as:

Z(1) = X(1)*Y(1)
Z(2) = X(1)*Y(2) + X(2)*Y(2)
Z(3) = X(1)*Y(3) + X(2)*Y(3) + X(3)*Y(3)
etc.

This can be written as:

$$Z_i = \sum_{j=1}^{i} X_{i-j+1} Y_j \qquad \textbf{(EQ 3-30)}$$

where i goes from 1 to n_x+n_y-1. This formula may look slightly different than the formulas in some references. This is accounted for by the fact that X is zero for indices greater than n_x and Y is zero for indices greater than n_y. Also, since DATAPLOT does not generate convolution via the fast Fourier transform, no zero padding is required.

SYNTAX

LET <z> = CONVOLUTION <y> <x> <SUBSET/EXCEPT/FOR qualification>

where <y> is the data stream variable with n_y elements;
<x> is the response variable with n_x elements;
<z> is a variable containing the computed convolution values (the length is n_x+n_y-1);
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES

LET Y3 = CONVOLUTION Y1 Y2

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

INTEGRAL	=	Compute a numerical integral of a variable or a function.
ROOT	=	Compute the roots of an equation.
DECONVOLUTION	=	Compute the discrete deconvolution of two variables.
FFT	=	Compute the Fast Fourier Transform of two variables.

REFERENCE

“Numerical Recipes: The Art of Scientific Computing (FORTRAN Version),” Press, Flannery, Teukolsky, and Vetterling, Cambridge University Press, 1989 (chapters 12 and 13).

APPLICATIONS

Mathematics

IMPLEMENTATION DATE

Pre-1987

PROGRAM

```
LET FUNCTION F1 = ((X+1)**2)/((X**2)+1)/2
LET FUNCTION F2 = 2**((-(X+1.3/2)**2)/(A**2))-2**((-(X-1.3/2)**2)/(A**2))
LET A = 0.85
.
LET XMIN = -7
LET XINC = .1
LET XMAX = 7
LET X = SEQUENCE XMIN XINC XMAX
.
LET Y1 = F1
LET Y2 = F2
LET Y3 = CONVOLUTION Y1 Y2
LET Y3 = Y3*XINC
.
LET X3MIN = 2*XMIN
LET X3MAX = 2*XMAX
LET X3 = SEQUENCE X3MIN XINC X3MAX
.
LINES SOLID SOLID DOTTED
TITLE DEMONSTRATE THE CONVOLUTION COMMAND
PLOT Y1 Y2 VS X AND
PLOT Y3 VS X3
```